
1 Overview

This document lists TTCN changes needed to correct problems in the TTCN implementation of test case 7.3.5.5 which is part of the LTE/SAE test suite ‘iwd-EUTRA-TVB2009-12_D10wk37’. 
In brief, the TC is reworked to introduce “timed sending”.

Corrections 
3.1 Change 1 /  Correction to Test Body of TC 7.3.5.5
	Function name
	f_EUTRA_7_3_5_5_TestBody

	Reason for change
	The test case expects PDCP PDUs in steps 13 and 14 to be received in separate PDUs, but not in a list. However, these PDUs may as well be provided in a list.


	Summary of change
	The reception of the PDCP PDUs to be looped back is handled in such a way that the PDUs are always to be received in a list. Toa achieve this UL Grant provision is inhibited and then restored.

Otherwise 2 types of reception (2 individual PDUs vs. reception in a list) would have to be handled.

New appropriate constraints are introduced.New appropriate constraints are introduced.

	TTCN module
	PDCP_Testcases.ttcn 

	MCC160 Comment
	


Before:

	  //@siclog "Step 11" siclog@
  // Check: Does the UE transmit PDCP Data PDU via RLC-AM RB with the following content back to the SS?
  // D/C field  = 1 (PDCP Data PDU) and PDCP SN=0. Data: previously received packet in PDCP Data PDU #0
  t_WatchDog.start;
  alt {
    [] DRB.receive (car_DRB_PDCP_PDU_Ind(eutra_Cell2, p_PDCP_Rec.RB_Id, ?))
      {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 11");
      }
    [] t_WatchDog.timeout
      {
      }
  }; // alt
  //@siclog "Step 12" siclog@
  // The SS sends the PDCP Data PDU#1 via RLC-AM RB with the following content to the UE:
  // PDCP Data PDU #1(D/C field  = 1 (PDCP Data PDU) and PDCP SN=1)
  f_UPSend(p_PDCP_Rec, eutra_Cell2, v_Data2);
  //@siclog "Step 13" siclog@
  // Check: Does the UE transmit a PDCP Data PDU via RLC-AM RB with the following content back to the SS?
  // D/C field  = 1 (PDCP Data PDU) and PDCP SN=1. Data: previously received packet in PDCP Data PDU #1
  f_UPRcv (p_PDCP_Rec, eutra_Cell2, v_Data2);
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");
  //@siclog "Step 14" siclog@
  // Check: Does the UE transmit PDCP Data PDU via RLC-AM RB with the following content back to the SS?
  // D/C field  = 1 (PDCP Data PDU) and PDCP SN=2, data: previously received packet in PDCP Data PDU #2
  f_UPRcv (p_PDCP_Rec, eutra_Cell2, v_Data3);
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 14");



After:

	  // inhibit loopback
  f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell2, cs_TimingInfo_Now);
  //@siclog "Step 12" siclog@
  // The SS sends the PDCP Data PDU#1 via RLC-AM RB with the following content to the UE:
  // PDCP Data PDU #1(D/C field  = 1 (PDCP Data PDU) and PDCP SN=1)
  f_UPSend(p_PDCP_Rec, eutra_Cell2, v_Data2);
  //@siclog "Step 13" siclog@
  // Check: Does the UE transmit a PDCP Data PDU via RLC-AM RB with the following content back to the SS?
  // D/C field  = 1 (PDCP Data PDU) and PDCP SN=1. Data: previously received packet in PDCP Data PDU #1
  //@siclog "Step 14" siclog@
  // Check: Does the UE transmit PDCP Data PDU via RLC-AM RB with the following content back to the SS?
  // D/C field  = 1 (PDCP Data PDU) and PDCP SN=2, data: previously received packet in PDCP Data PDU #2
  f_EUTRA_StartDefULGrantTransmission(eutra_Cell2, cs_TimingInfo_Now);  
  f_UPRcv2PDUs (p_PDCP_Rec, eutra_Cell2, v_Data2, v_Data3);
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 13");
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 14");



New function to receive 2 PDCP PDUs in a list:

	function f_UPRcv2PDUs (inout PDCP_SS_State_Type p_PDCP_Rec,
                       CellId_Type p_CellId,
                       template (present) octetstring p_PdcpSdu1,
                       template (present) octetstring p_PdcpSdu2)
  runs on EUTRA_PTC
{
  var integer v_Next_PDCP_RX_SN := p_PDCP_Rec.Next_PDCP_RX_SN;
  var PDCP_SDU_Type v_ReceivedSdu1, v_ReceivedSdu2, v_ReceivedSdu3;
  var octetstring v_DecipheredSdu1, v_DecipheredSdu2, v_DecipheredSdu3;
  var DRB_COMMON_IND v_ReceivedPdu;
  // for the time being RX_HFN will always be 0, so Next_PDCP_RX_SN is used instead of RX_COUNT !!!
  // Build SN template to be used
  if (p_PDCP_Rec.PDCP_SN_Size != PDCP_SNLength12)
  {
      FatalError(__FILE__, __LINE__, "PDCP_SN_Size incorrect");
  }
  DRB.receive (car_DRB1_PDCP_DataListLongSN_2(p_CellId,
                                    cr_PDCP_DataLongSN_2(int2bit(v_Next_PDCP_RX_SN, 12), p_PdcpSdu1),
                                    cr_PDCP_DataLongSN_2(int2bit((v_Next_PDCP_RX_SN + 1), 12), p_PdcpSdu2)))-> value v_ReceivedPdu;
  // Check payload if required
  v_ReceivedSdu1 := v_ReceivedPdu.U_Plane.SubframeData.PduSduList.PdcpPdu[0].DataLongSN.SDU;
  v_DecipheredSdu1 := fx_AsDeciphering (v_ReceivedSdu1,
                                         p_PDCP_Rec.CipheringAlgorithm,
                                         p_PDCP_Rec.KUPenc,
                                         int2oct(p_PDCP_Rec.Next_PDCP_RX_SN, 4),
                                         int2bit(p_PDCP_Rec.RB_Id.Drb - 1, 5));   /* R5-101050: BearerId changed to 5 bits */ 
  if (not match(v_DecipheredSdu1, p_PdcpSdu1))
  {
    f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Incorrect PDCP SDU");
  }
  if (p_PDCP_Rec.Next_PDCP_RX_SN == p_PDCP_Rec.Maximum_PDCP_SN)
    {
      p_PDCP_Rec.Next_PDCP_RX_SN := 0;
      p_PDCP_Rec.RX_HFN := p_PDCP_Rec.RX_HFN + 1; // currently not used further
    }
  else
    {
      p_PDCP_Rec.Next_PDCP_RX_SN := p_PDCP_Rec.Next_PDCP_RX_SN + 1;
    }
  v_ReceivedSdu2 := v_ReceivedPdu.U_Plane.SubframeData.PduSduList.PdcpPdu[1].DataLongSN.SDU;
  v_DecipheredSdu2 := fx_AsDeciphering (v_ReceivedSdu2,
                                         p_PDCP_Rec.CipheringAlgorithm,
                                         p_PDCP_Rec.KUPenc,
                                         int2oct(p_PDCP_Rec.Next_PDCP_RX_SN, 4),
                                         int2bit(p_PDCP_Rec.RB_Id.Drb - 1, 5));   /* R5-101050: BearerId changed to 5 bits */ 
  if (not match(v_DecipheredSdu2, p_PdcpSdu2))
  {
    f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Incorrect PDCP SDU");
  }
  if (p_PDCP_Rec.Next_PDCP_RX_SN == p_PDCP_Rec.Maximum_PDCP_SN)
    {
      p_PDCP_Rec.Next_PDCP_RX_SN := 0;
      p_PDCP_Rec.RX_HFN := p_PDCP_Rec.RX_HFN + 1; // currently not used further
    }
  else
    {
      p_PDCP_Rec.Next_PDCP_RX_SN := p_PDCP_Rec.Next_PDCP_RX_SN + 1;
    }
} // f_UPRcv2PDUs



New ASP constraint to receive 2 PDCP PDUs in a list:
	//ASP Constraint
template DRB_COMMON_IND car_DRB1_PDCP_DataListLongSN_2(CellId_Type p_CellId,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN1,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN2) :=
{ /* @status     */
Common := cr_IndAspCommonPart_RB (p_CellId, tsc_RbId_DRB1),
  U_Plane := {
    SubframeData := { 
      PduSduList := {
        PdcpPdu := {
          {DataLongSN := p_PDCP_PDULongSN1},
          {DataLongSN := p_PDCP_PDULongSN2}          
        }
      },
      NoOfTTIs := ?
    } 
  } // U_Plane
}; // car_DRB1_PDCP_DataListLongSN_2



New constraint  to receive a PDCP PDUs of PDCP_DataPdu_LongSN_Type :

	template PDCP_DataPdu_LongSN_Type cr_PDCP_DataLongSN_2 ( template (present) B12_Type p_SN,
                                            template (present) octetstring p_PdcpSdu) :=
{ /* @desc Template used for receiving a PDCP PDU with long SN */
    D_C := tsc_D_C_DataPDU,
    Reserved := tsc_R_Bit & tsc_R_Bit & tsc_R_Bit,
    SequenceNumber := p_SN,
    SDU := p_PdcpSdu
};


4 References

	-
	n/a

